Pensieve header: A 4D algebra whose associated knot invariant is Alexander.

In[•]:=
```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Classes\\21-1350-KnotTheory"];
<< Common.m
```

```
Loading KnotTheory` version of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.
```

In[•]:=

$$MT = \begin{pmatrix} & a & b & c & d \\ a & a & b & 0 & 0 \\ b & 0 & 0 & a & b \\ c & c & d & 0 & 0 \\ d & 0 & 0 & c & d \end{pmatrix};$$

```
ℰ_ // m_{i_,j_→k_} := Expand[ℰ] /. Flatten@
    Table[MT⟦α, 1⟧_i MT⟦1, β⟧_j → (MT⟦α, β⟧ /. v : (a | b | c | d) ⧴ v_k), {α, 2, 5}, {β, 2, 5}];
η_{i_} :=
  a_i +
   d_i;
```

In[•]:=
```
KBasis[i_] := {a_i, b_i, c_i, d_i};
KBasis[i_, is__] := Flatten@Outer[Times, KBasis[i], KBasis[is]]
```

In[•]:=

$$R_{i\_,j\_} := T^{-1/2}\left(a_i a_j + T a_i d_j + (1-T)\, b_j c_i + d_i a_j - T d_i d_j\right);$$
$$\overline{R}_{i\_,j\_} := T^{1/2}\left(a_i a_j + T^{-1} a_i d_j + \left(1-T^{-1}\right) b_j c_i + d_i a_j - T^{-1} d_i d_j\right);$$
$$C_{i\_} := T^{-1/2}\,(a_i - d_i);\ \overline{C}_{i\_} := T^{1/2}\,(a_i - d_i);$$

RVK and Z

RVK, rot, Z modified from 2016-09/OneSmidgen.nb. See also in AP/Projects/SL2Invariant/.

Some details of the code below are at http://drorbn.net/bbs/show?shot=Dror-160920-151350.jpg.

In[•]:=
```
RVK::usage =
  "RVK[xs, rots] represents a Rotational Virtual Knot with a list of n Xp/Xm crossings
    xs and a length 2n list of rotation numbers rots. Crossing
    sites are indexed 1 through 2n, and rots⟦k⟧ is the rotation
    between site k-1 and site k. RVK is also a casting operator
    converting to the RVK presentation from other knot presentations.";
```

In[∘]:=
```
RVK[pd_PD] := Module[{n, xs, x, rots, front = {1}, k},
   n = Length@pd; rots = Table[0, {2 n}];
   xs = Cases[pd, x_X :→ { Xp[x[[4]], x[[1]]]  PositiveQ@x
                           Xm[x[[2]], x[[1]]]  True         ];
   For[k = 1, k ≤ 2 n, ++k,
    If[FreeQ[front, -k],
     front = Flatten@Replace[front, k → (xs /. {
            Xp[k, l_] | Xm[l_, k] :→ {l + 1, k + 1, -l},
            Xp[l_, k] | Xm[k, l_] :→ (++rots[[l]]; {-l, k + 1, l + 1}),
            _Xp | _Xm :→ {}
           }), {1}],
     Cases[front, k | -k] /. {k, -k} :→ --rots[[k]];
    ]
   ];
   RVK[xs, rots] ];
RVK[K_] := RVK[PD[K]];
```

In[∘]:=
```
rvk = RVK[Knot[6, 3]]
```

KnotTheory: Loading precomputed data in PD4Knots`.

Out[∘]= RVK[{Xp[1, 4], Xp[3, 8], Xm[9, 12], Xm[5, 10], Xm[11, 6], Xp[7, 2]}, {0, 0, 0, -1, 0, 0, 0, 0, 0, 1, 0, 0}]

In[∘]:= $R_{1,2}$

Out[∘]=
$$\frac{a_1 a_2 + (1 - T)\, b_2 c_1 + a_2 d_1 + T a_1 d_2 - T d_1 d_2}{\sqrt{T}}$$

In[∘]:= C_1

Out[∘]=
$$\frac{a_1 - d_1}{\sqrt{T}}$$

In[∘]:=
$$\text{rot}_{i\_}[n\_] := \text{rot}_i[n] = \begin{cases} \eta_i & n == 0 \\ C_\$ \,\text{rot}_i[n-1] \mathbin{//} m_{i,\$\to i} & n > 0 \\ \overline{C}_\$ \,\text{rot}_i[n+1] \mathbin{//} m_{i,\$\to i} & n < 0 \end{cases}$$

In[∘]:= $\text{rot}_1[15]$

Out[∘]=
$$\frac{a_1}{T^{15/2}} - \frac{d_1}{T^{15/2}}$$

In[∘]:=
```
z = Product[
   {i, j} = List @@ x;
   (x /. {_Xp ⇒ R_{i,j}, _Xm ⇒ R̄_{i,j}}) rot_{i-1/2}[rvk⟦2, i⟧] rot_{j-1/2}[rvk⟦2, j⟧] // m_{i-1/2,i→i} //
    m_{j-1/2,j→j},
   {x, rvk⟦1⟧}
  ]
```

Out[∘]=
$$(a_1 a_4 + b_4 c_1 - T b_4 c_1 + a_4 d_1 - T a_1 d_4 + T d_1 d_4) \left(\frac{a_2 a_7}{\sqrt{T}} + \frac{b_2 c_7}{\sqrt{T}} - \sqrt{T}\, b_2 c_7 + \sqrt{T}\, a_7 d_2 + \frac{a_2 d_7}{\sqrt{T}} - \sqrt{T}\, d_2 d_7\right)$$
$$\left(\frac{a_3 a_8}{\sqrt{T}} + \frac{b_8 c_3}{\sqrt{T}} - \sqrt{T}\, b_8 c_3 + \frac{a_8 d_3}{\sqrt{T}} + \sqrt{T}\, a_3 d_8 - \sqrt{T}\, d_3 d_8\right)$$
$$\left(a_5 a_{10} + b_{10} c_5 - \frac{b_{10} c_5}{T} + a_{10} d_5 - \frac{a_5 d_{10}}{T} + \frac{d_5 d_{10}}{T}\right)$$
$$\left(\sqrt{T}\, a_6 a_{11} - \frac{b_6 c_{11}}{\sqrt{T}} + \sqrt{T}\, b_6 c_{11} + \frac{a_{11} d_6}{\sqrt{T}} + \sqrt{T}\, a_6 d_{11} - \frac{d_6 d_{11}}{\sqrt{T}}\right)$$
$$\left(\sqrt{T}\, a_9 a_{12} - \frac{b_{12} c_9}{\sqrt{T}} + \sqrt{T}\, b_{12} c_9 + \sqrt{T}\, a_{12} d_9 + \frac{a_9 d_{12}}{\sqrt{T}} - \frac{d_9 d_{12}}{\sqrt{T}}\right)$$

In[∘]:=
```
Short[Expand[z], 10]
```

Out[∘]//Short=
$$a_1 a_2 a_3 a_4 a_5 a_6 a_7 a_8 a_9 a_{10} a_{11} a_{12} + a_2 a_3 a_5 a_6 a_7 a_8 a_9 a_{10} a_{11} a_{12} b_4 c_1 - T a_2 a_3 a_5 a_6 a_7 a_8 a_9 a_{10} a_{11} a_{12} b_4 c_1 + \ll 51544 \gg + d_1 d_2 d_3 d_4 d_5 d_6 d_7 d_8 d_9 d_{10} d_{11} d_{12}$$

In[∘]:=
```
{6^6, Length[Expand[z]]}
```

Out[∘]= {46 656, 34 375}

In[∘]:=
```
Do[z = m_{1,i→1}[z], {i, 2, 12}];
Factor@z
```

Out[∘]=
$$\frac{(1 - 3 T + 5 T^2 - 3 T^3 + T^4)\,(a_1 + d_1)}{T^2}$$

In[∘]:=
```
Alexander[Knot[6, 3]][T]
```

Out[∘]=
$$5 + \frac{1}{T^2} - \frac{3}{T} - 3 T + T^2$$

In[∘]:=
```
Z[K_] := Z[RVK@K];
Z[rvk_RVK] := Module[{ζ, done, st, c, χ, i, j, k},
  ζ = 1; done = {}; st = Range[2 Length[rvk[[1]]]];
  Do[
   {i, j} = List @@ c;
   χ = c /. {_Xp :→ R_{i,j}, _Xm :→ R̄_{i,j}};
   Do[χ = (rot_θ[rvk[[2, k]]] χ) // m_{θ,k→k}, {k, {i, j}}];
   ζ *= χ;
   Do[
    If[MemberQ[done, k + 1], ζ = ζ // m_{k,k+1→k}; st = st /. k + 1 → k];
    If[MemberQ[done, k - 1], ζ = ζ // m_{st[[k-1]],k→st[[k-1]]}; st = st /. k → st[[k - 1]]],
    {k, {i, j}}];
   done = done ⋃ {i, j},
   {c, rvk[[1]]}
  ];
  Factor@ζ
 ]
```

In[∘]:=
```
K = Knot[8, 17]; Factor@Alexander[K][T]
z = Z[K]
```

Out[∘]= $-\frac{1-4T+8T^2-11T^3+8T^4-4T^5+T^6}{T^3}$

Out[∘]= $-\frac{\left(1-4T+8T^2-11T^3+8T^4-4T^5+T^6\right)(a_1+d_1)}{T^3}$

In[∘]:=
```
Timing@Table[Simplify[Alexander[K][T] η₁ == Z[K]], {K, AllKnots[{3, 9}]}]
```

Out[∘]= {19.3125, {True, True}}

In[∘]:=
```
ZF[K_] := Z@ThinPosition@K;
```

In[∘]:=
```
Timing@Union@Table[Simplify[Alexander[K][T] η₁ == ZF[K]], {K, AllKnots[{3, 9}]}]
```

Out[∘]= {4.01563, {True}}

In[∘]:=
```
Timing@Union@Table[Simplify[Alexander[K][T] η₁ == ZF[K]], {K, AllKnots[{3, 10}]}]
```

In[∘]:=
```
Timing[ZF[GST48]]
```

Out[∘]= $\left\{38.1094, -\frac{\left(-1+2T-T^2-T^3+2T^4-T^5+T^8\right)\times\left(-1+T^3-2T^4+T^5+T^6-2T^7+T^8\right)(a_1+d_1)}{T^8}\right\}$

In[ ]:= **ZF /@ RibbonKnots**

Out[ ]=
$$\left\{-\frac{(-2+T)\times(-1+2T)\ (a_1+d_1)}{T},\ \frac{(2-2T+T^2)\times(1-2T+2T^2)\ (a_1+d_1)}{T^2},\right.$$
$$-\frac{(-1+T-2T^2+T^3)\times(-1+2T-T^2+T^3)\ (a_1+d_1)}{T^3},\ \frac{(1-T+T^2)^2\ (a_1+d_1)}{T^2},$$
$$-\frac{(-1+2T-3T^2+T^3)\times(-1+3T-2T^2+T^3)\ (a_1+d_1)}{T^3},$$
$$\frac{(3-3T+T^2)\times(1-3T+3T^2)\ (a_1+d_1)}{T^2},\ -\frac{(-2+T)\times(-1+2T)\ (a_1+d_1)}{T},$$
$$-\frac{(-3+2T)\times(-2+3T)\ (a_1+d_1)}{T},\ -\frac{(-2+2T-2T^2+T^3)\times(-1+2T-2T^2+2T^3)\ (a_1+d_1)}{T^3},$$
$$\frac{(2-4T+T^2)\times(1-4T+2T^2)\ (a_1+d_1)}{T^2},\ -\frac{(-1+3T-4T^2+T^3)\times(-1+4T-3T^2+T^3)\ (a_1+d_1)}{T^3},$$
$$\frac{(1-T+2T^2-2T^3+T^4)\times(1-2T+2T^2-T^3+T^4)\ (a_1+d_1)}{T^4},$$
$$-\frac{(-1+3T-4T^2+T^3)\times(-1+4T-3T^2+T^3)\ (a_1+d_1)}{T^3},\ -\frac{(-2+T)\times(-1+2T)\ (1-T+T^2)^2\ (a_1+d_1)}{T^3},$$
$$\frac{(1-T+T^2)^4\ (a_1+d_1)}{T^4},\ \frac{(1-3T+3T^2-3T^3+T^4)^2\ (a_1+d_1)}{T^4},$$
$$\frac{(2-2T+T^2)\times(1-2T+2T^2)\ (a_1+d_1)}{T^2},\ \frac{(1-3T+T^2)^2\ (a_1+d_1)}{T^2},\ \frac{(1-T+T^2)^2\ (a_1+d_1)}{T^2},$$
$$\left.\frac{(1-T+T^3)\times(1-T^2+T^3)\ (a_1+d_1)}{T^3},\ -\frac{(-1+T-2T^2+T^3)\times(-1+2T-T^2+T^3)\ (a_1+d_1)}{T^3}\right\}$$